Программа mode64.bat является учебной и кратчайшим образом переводит процессор в 64-битный режим, если он поддерживается. В этом режиме при запрещенных прерываниях производится вывод строки на экран в результате прямой записи в видеопамять. Для запуска программы необходимо перезагрузить компьютер в среде MS-DOS. Для этого можно, например, отформатировать и сделать загрузочной дискету, перенести на нее исполняемый файл mode64.exe и установить в BIOS значение «Floppy» для первого загрузочного устройства. При написании программы были использованы руководства
· AMD64 Architecture Programmer’s Manual. Volume 2: System Programming. Publication No. 24593
· AMD64 Architecture Programmer’s Manual. Volume 3: General-Purpose and System Instructions. Publication No. 24594

размещенные на www.amd.com. Аналогичные руководства по архитектуре Intel-64 расположены на developer.intel.com/products/processor/manuals/index.htm:

· Intel® 64 and IA-32 Architectures Software Developer’s Manual. Volume 3A: System Programming Guide, Part 1. Order Number: 253668
· Intel® 64 and IA-32 Architectures Software Developer’s Manual. Volume 2A: Instruction Set Reference, A-M. Order Number: 253666
· Intel® 64 and IA-32 Architectures Software Developer’s Manual. Volume 2B: Instruction Set Reference N-Z. Order Number: 253667
Информацию о переходе в 64-битный режим и соответствующие исходные тексты можно также найти в статьях «Исследовательская работа № 1» и «Исследовательская работа № 2», размещенных на www.xakep.ru. Я узнал о них после написания программы. Предполагается, что читатель имеет общее представление о 64-разрядной архитектуре, а также о работе процессора в защищенном режиме.
Порядок смены режимов таков: реальный ( защищенный ( режим совместимости ( 64-битный ( режим совместимости ( защищенный ( реальный. Выделенные курсивом режимы относятся к так называемому длинному режиму, являющемуся расширением защищенного режима.
Ниже приводятся комментарии к программе.
«Самотрансляция»
; @echo off
; tasm /m2 mode64.bat

; pause

; if errorlevel 1 goto exit

; tlink /32 /x  mode64.obj

; del mode64.obj

; pause

; goto exit

При запуске mode64.bat командный интерпретатор игнорирует начальные точки с запятой. По окончании работы он переходит на метку exit в конце программы. В свою очередь, при трансляции (вторая строка) ассемблер игнорирует указанные выше закомментированные строки.
Для трансляции и компиляции программы использовались Turbo Assembler версии 4.1 и Turbo Link версии 4.0. Опции: /m2 — два прохода транслятора для реализации переходов «вперед», /32 — разрешение 32-битных смещений, /x — подавление генерации map-файла.
Заголовок
.586P

assume
cs:cseg,ds:dseg,ss:sseg

sseg

segment stack 'stack' use16



db 80h dup (0)

sseg

ends

Разрешаются привилегированные инструкции. Размер сегмента стека выбран 80h байт.
Инициализация системных структур в реальном режиме

cseg

segment use16

linear
proc



movzx eax,bx



shl eax,4



ret

linear
endp

Небольшая процедура помещает сегментный адрес (регистр BX) в EAX и умножает на 16, преобразуя его в линейный адрес сегмента.
er:

lea dx,info



mov ah,9



int 21h



mov dx,si



mov ah,9



int 21h



lea dx,press



mov ah,9



int 21h



mov ah,8



int 21h



jmp ex

; на выход

Ветвь программы, выводящая сообщения при ошибках. Вначале выводится описание программы — строка info (см. ниже сегмент данных). Затем в зависимости от значения регистра SI либо строка notsupport о том, что процессор не поддерживает 64 бита, либо строка bootmess о том, что требуется запуск программы в MS-DOS, а не в Windows. Наконец, выводится строка press о необходимости нажатия любой клавиши. Функция 8h считывает символ с ожиданием и без эха. Затем на выход.
Точка входа. Начинаем с проверки типа процессора.
start:
mov ax,dseg



mov ds,ax



lea si,notsupport



push sp



pop ax



cmp ax,sp



jne short er

Вряд ли эта программа когда-либо будет запущена на 86-м. Тем не менее, отмечаем, что этот процессор при выполнении PUSH SP вначале уменьшает SP на 2, потом помещает его в стек, но не наоборот, что его и выдает. Используем короткую форму перехода на метку er, иначе ассемблер может вставить 386-ю инструкцию перехода, непосильную для 86-го.


mov bx,0fff0h



smsw ax



and ax,bx



cmp ax,bx



je short er

Пропускаем такую экзотику как 186-е и переходим к 286-м. Они еще не доросли до регистра управления CR0 и используют 16-битное слово состояния машины MSW. Выполняя SMSW AX, возвращают в AX единичные биты с 4-го по 15-й в отличие от более продвинутых процессоров.


pushfd



pop eax



mov ebx,eax



btc eax,21
; пробуем менять флаг ID



push eax



popfd



pushfd



pop eax

;

push ebx

;

popfd

; можно подтереть там, где нагадили



cmp ebx,eax



je er

Вышеуказанная часть кода выполнится на 386+. Возможность менять флаг идентификации ID (21-й бит регистра EFLAGS) — показатель еще большей продвинутости, соответствующий 486+ процессорам, поддерживающим инструкцию CPUID.


mov eax,80000000h



mov edi,eax



inc di



cpuid



cmp eax,edi



jb er

Расширенная функция 80000000h инструкции CPUID возвращает в EAX максимальный номер расширенных функций CPUID. Нам нужна 80000001h-я функция, возвращающая в ECX и EDX специфическую информацию о процессоре.


mov al,1
; ф-ция 80000001h



cpuid



bt edx,29



jnc er

Проверяем 29-й бит в регистре EDX, единичное значение которого указывает на поддержку архитектуры-64.


lea si,bootmess



pushfd



pop eax



bt eax,17



jc er

Последняя проверка — запущена ли программа в MS-DOS (реальный режим) или в Windows (режим V8086, т.к. программа досовская). Для этого анализируется бит виртуального режима VM (17-й) регистра EFLAGS. Проверка выполняется ввиду невозможности выполнения нижележащего кода в Windows с наименьшими привилегиями 3-го уровня.
Заполняем системные структуры.


mov bx,ds



call linear



mov addrgdt,eax

Вычисляем линейный адрес сегмента данных, в начале которого расположена таблица глобальных дескрипторов GDT (см. ниже). Значение заносится в переменную addrgdt, используемую далее при загрузке регистра GDTR. Приступаем к формированию таблиц страниц. Страничное преобразование — обязательное условие входа в длинный режим.


and bx,0ff00h



add bx,200h



call linear



mov cr3,eax

Округляем сегментный адрес на границу 4096-байтной страницы, добавляем 200h, вычисляем соответствующий линейный адрес. Таким образом, на расстоянии несколько меньшем, чем 2000h от конца программы будет располагаться таблица карт страниц уровня 4 для расширенных физических адресов (PAE) длинного режима. Ее адрес заносим в регистр управления CR3.


push ds



mov ds,bx



xor edi,edi  ; для установки нулевых старших битов 32 – 63 адресов



xor si,si



mov al,7

Настраиваем сегментный регистр DS для адресации таблиц, сохранив в стеке его старое значение. Механизм трансляции линейных адресов в физические в длинном режиме сложнее чем в защищенном режиме. При использовании страниц размером 4 кб таблицы состоят из 8-байтных входов, в которых бит 0 — бит присутствия/отсутствия страницы, бит 1 — бит записи/только чтения, бит 2 — бит пользователя/супервизора. Эти биты будем устанавливать в 1, поместив в регистр AL значение 111b = 7. Описание битов 3-11 здесь не приводим, их устанавливаем в ноль. Биты 12-63 соответствуют таким же по номеру битам физических адресов таблиц нижележащего уровня. Биты 12-31 входов заполняем с помощью EAX, а старшие биты 32-63 очевидно будут нулевыми. Для их заполнения используем установленный в ноль регистр EDI. Всего уровней четыре — таблица карт страниц уровня 4, таблицы указателей каталогов страниц, таблицы каталогов страниц и таблицы страниц. При преобразованиях линейных адресов в физические биты 39-47 линейного адреса определяют номер входа в таблице карт страниц уровня 4, биты 30-38 — номер входа в таблице указателей каталогов страниц, биты 21-29 — номер входа в таблице каталогов страниц, биты 12-20 — номер входа в таблице страниц и, наконец, биты 0-11 — смещение от начала физической страницы. С помощью регистра SI указываем смещение от начала таблиц.
Будем заполнять таблицы так, чтобы они соответствовали нижнему мегабайту адресного пространства и физические адреса совпадали с линейными.


mov cx,3

again1:
add eax,1000h



mov [si],eax



mov [si+4],edi



add si,1000h



loop again1

Для этого достаточно подготовить только по одному входу в таблице карт страниц уровня 4, в таблице указателей каталогов страниц и в таблице каталогов страниц, расположив эти входы в начале каждой из трех соседних 4096-байтных страниц.


mov cx,100h



xor eax,eax
; отображение один к одному



mov al,7

again2:
mov [si],eax



mov [si+4],edi



add eax,1000h



add si,8



loop again2



pop ds

; таблицы готовы!

В следующей странице помещаем 1 Mб / 4 кб = 256 = 100h входов таблицы страниц. Во входах указываем физические адреса страниц, начиная с нулевого адреса и заканчивая 0FF000h с шагом 1000h. Таблицы заполнены.
Остается рассчитать линейные адреса текущего сегмента кода cseg, сегмента кода защищенного режима cseg32, сегмента кода 64-битного режима cseg64 и линейные адреса точек возврата.


mov bx,cs



mov segaddr,bx



call linear



mov lowbase,ax



shr eax,16



mov middbase,al

Сегментный адрес текущего кода помещается в переменную segaddr. Кроме того, младшую (биты 0-15) и среднюю (биты 16-23) части соответствующего линейного адреса заносим в поля базы дескриптора 16-разрядного кода. Таблица глобальных дескрипторов заполнена, при этом селектор 8h соответствует 32-разрядному коду, селектор 10h — 64-разрядному коду, а селектор 18h — 16-разрядному коду.


mov bx,cseg32



call linear



mov addr32,eax


При переключении в защищенный режим потребуется линейный адрес 32-разрядного кода, который сохраняем в переменной addr32 в текущем сегменте.



assume ds:cseg32



push ds



push eax



mov ds,bx
Адресуемся к 32-разрядному сегменту и сохраняем в стеке сегментный адрес dseg и линейный адрес cseg32.



mov bx,cseg64



call linear



mov addr64,eax
Вычисляем линейный адрес 64-разрядного кода и помещаем его в переменную addr64, определенную в сегменте cseg32.


pop eax



add eax,offset ret32



mov ds,bx



assume ds:cseg64



mov to32,eax
Находим адрес точки возрата ret32 в 32-разрядный код из 64-битного режима, адресуемся к сегменту cseg64 и записываем этот адрес в переменную to32 в сегменте cseg64.



pop ds



assume ds:dseg
Восстанавливаем адресацию сегмента данных. Инициализация системных структур завершена.


cli



lgdt pdescr



mov eax,cr0



or al,1



mov cr0,eax
Запрещаем аппаратные прерывания и далее до возврата в реальный режим их не обрабатываем, загружаем регистр GDTR, устанавливаем бит 0 в регистре CR0. Процессор переведен в защищенный режим.


db 66h



db 0eah

; JMP FAR
addr32
dd ?



dw 8
Выполняем прямой переход JMP FAR на 32-разрядный код с селектором CS = 8h по смещению EIP = addr32, вычисленному ранее. Используем префикс 66h замены размера операнда, поскольку у инструкции JMP FAR, расположенной в 16-разрядном сегменте, 32-разрядный операнд addr32.
Точка возврата из 32-разрядного кода.
ret16:
mov ax,cr0



and al,0feh



mov cr0,eax

Сбрасываем бит 0 в регистре CR0 и переводим процессор в реальный режим.


db 0eah

; JMP FAR


dw offset return

segaddr
dw ?

Селекторы защищенного режима недействительны в реальном режиме, поэтому, выполняя инструкцию дальнего перехода, загружаем CS сегментным адресом.
return:
sti

ex:

mov ax,4c00h



int 21h

cseg

ends

Разрешаем аппаратные прерывания и завершаем работу программы.

32-разрядный код
В защищенном режиме не используем сегментные регистры DS, ES и SS. Поэтому в таблице глобальных дескрипторов не предусматривается описание сегментов данных и стека, а в указанных регистрах остаются «теневые» значения реального режима.
Готовим переход в длинный режим.
cseg32
segment use32



mov ax,cr4



bts eax,5



mov cr4,eax

Установка бита 5 в регистре CR4 разрешает расширенные физические адреса.


mov ecx,0c0000080h



rdmsr



bts eax,8



wrmsr

Сохраняем в паре EDX:EAX машинно-специфичный регистр (MSR) с номером C0000080h — регистр расширенных возможностей. Биты 8 и 10 этого регистра ответственны за работу процессора в длинном режиме. Бит 10 (только для чтения) устанавливается аппаратно и показывает, находится ли процессор в данном режиме. Установка бита 8 лишь разрешает режим, но не переводит в него процессор. Устанавливаем этот бит и записываем значение из EDX:EAX в MSR.


mov eax,cr0



bts eax,31



mov cr0,eax

Только включение страничного преобразования переводит процессор в длинный режим (бит 31 регистра CR0). При этом следующей инструкцией должна быть инструкция перехода. Вспомним, что таблицы страниц были заблаговременно подготовлены, а регистр CR3 инициализирован. На самом деле в данный момент процессор находится в режиме совместимости и выполняет 32-разрядный код, поскольку в CS находится селектор 8h этого кода. Но обработка прерываний, если бы они были разрешены, происходила бы по правилам длинного режима.


db 0eah

; JMP FAR

addr64
dd ?



dw 10h

Дальний переход на 64-разрядный код, описываемый селектором 10h, окончательно переводит процессор в 64-битный режим!
Точка возврата из 64-разрядного кода. При возврате процессор находится в режиме совместимости с CS = 8h.
ret32

label word



mov ax,cr0



btr eax,31



mov cr0,eax

Отключаем страничное преобразование и тем самым переводим процессор в обычный защищенный режим.


mov ecx,0c0000080h



rdmsr



btr eax,8



wrmsr

Запрещаем возможность включения длинного режима.


mov eax,cr4



btr eax,5



mov cr4,eax

Запрещаем использовать расширенные физические адреса. Теперь нужно готовиться к возврату в реальный режим.


db 66h



db 0eah

; JMP FAR



dw offset ret16



dw 18h

cseg32
ends

Выполняем дальний переход на 16-разрядный код, описываемый селектором 18h. Дескриптор сегмента этого кода подготовлен согласно правилам реального режима — код разрешено читать, лимит сегмента — FFFFh. Префикс 66h замены размера операнда используется потому, что у инструкции JMP FAR в 32-разрядном сегменте операнд OFFSET RET16 — 16-разрядный.
Вывод строки на экран с помощью 64-разрядного кода

Полезным результатом работы следующего фрагмента кода будет вывод сообщения «Привет из 64-битного режима!» в начало 24-й строки экрана. Достигается это прямой записью в текстовую видеопамять, которая начинается, как известно, по адресу B8000h. Адрес памяти, соответствующий началу 24-й строки, равен B8000h + 24 * 80 * 2 = B8000h + F00h = B8F00h. Здесь учтено, что в строке 80 позиций, а на каждый символ приходится по два байта — код символа (естественно, в DOS-кодировке) и атрибут символа. Последний принимаем равным 7h — белый цвет, черный фон.
64-битный режим практически несовместим с защищенным режимом. Регистры общего назначения EAX, …, EDX, ESI, EDI, EBP, ESP расширены до 8-байтных RAX, …, RDX, RSI, RDI, RBP, RSP, и добавлены регистры R8 – R15. Системные шлюзы вызова, ловушек и прерываний расширены до 16 байт. Отменено переключение задач, отсутствует режим V8086, используется плоская модель памяти, частично используются теневые части сегментных регистров CS, FS и GS. Машинный код обоих режимов мало совместим. Есть и другие различия, о которых здесь не упоминаем.
Ясно, что для компилятора TASM непосильны 64-разрядные инструкции, поэтому сегмент cseg64 будем писать в машинных кодах, используя директивы определения данных DB, DW и DD. Сделать это нетрудно, поскольку для пересылки данных воспользуемся только инструкциями MOV RAX, imm64 и MOV moffset64, RAX, где imm64 — непосредственный 64-разрядный операнд, moffset64 — абсолютный 64-разрядный адрес. В последней инструкции вторым операндом может быть только аккумулятор AL, AX, EAX или RAX.
По умолчанию в 64-битном режиме используется 64-разрядная адресация и 32-разрядные данные. Для изменения разрядности данных до 64-х машинные инструкции предваряются однобайтными REX-префиксами, принимающими значения от 40h до 4Fh. Для нас представляет интерес префикс 48h с установленным битом 3, указывающим на то, что разрядность операнда равна 64.
cseg64
segment



db 48h



db 0b8h

; mov RAX, imm64


db 'П',7,'р',7,'и',7,'в',7

В регистр RAX пересылаются символы «П», «р», «и», «в» с атрибутами 7h.



db 48h



db 0a3h

; mov moffset64, RAX



dd 0b8f00h,0
; 64-разрядный адрес памяти
Содержимое регистра RAX пересылается в память по адресу B8F00h. При этом совершенно не важно, что находится в сегментном регистре DS (там остались данные реального режима). База сегмента данных в 64-битном режиме всегда равна 0h, а лимит вообще игнорируется.


db 48h,0b8h



db 'е',7,'т',7,' ',7,'и',7



db 48h,0a3h



dd 0b8f08h,0

Следующие восемь байтов пересылаются по адресу B8F08h.



db 48h,0b8h



db 'з',7,' ',7,'6',7,'4',7



db 48h,0a3h



dd 0b8f10h,0



db 48h,0b8h



db '-',7,'б',7,'и',7,'т',7



db 48h,0a3h



dd 0b8f18h,0



db 48h,0b8h



db 'н',7,'о',7,'г',7,'о',7



db 48h,0a3h



dd 0b8f20h,0



db 48h,0b8h



db ' ',7,'р',7,'е',7,'ж',7



db 48h,0a3h



dd 0b8f28h,0



db 48h,0b8h



db 'и',7,'м',7,'а',7,'!',7



db 48h,0a3h



dd 0b8f30h,0

Сообщение выведено на экран.

Теперь следует вернуться в 32-разрядный код, поместив в регистр CS с помощью инструкции дальнего перехода селектор 8h этого кода. Однако в 64-битном режиме в отличие от реального и защищенного отсутствует инструкция прямого перехода с кодом EAh. Используем инструкцию косвенного перехода.


db 0ffh,2dh
; JMP FWORD PTR to32



dd 0

Странным может показаться наличие нулевого двойного слова, следующего за инструкцией перехода FFh, 2Dh. В защищенном режиме в нем должен находиться 32-разрядный адрес переменной типа FWORD, в которой и указывается адрес перехода в виде CS:EIP. Однако в 64-битном режиме используется не абсолютная, а относительная адресация (еще одно отличие от защищенного режима). Это означает, что указанная переменная типа FWORD смещена относительно инструкции перехода на расстояние 0h, т.е. в данном случае непосредственно следует за ней.
to32

dd ?



dw 8

cseg64
ends

В эту переменную на этапе инициализации был записан адрес точки возврата в 32-разрядный код. Выполнив дальний переход, процессор переходит в режим совместимости с CS = 8h.
Сегмент данных
В начале сегмента размещаем таблицу глобальных дескрипторов, содержащую только дескрипторы кода. Такие дескрипторы для защищенного режима имеют следующую структуру:

	31               24
	23
	22
	21
	20
	19                  16
	15
	14  13
	12
	11
	10
	9
	8
	7                   0
	

	Базовый

адрес 31-24
	G
	D
	
	A
V

L
	Лимит 19-16
	P
	DPL
	1
	1
	C
	R
	A
	Базовый

адрес 23-16
	+4

	Базовый адрес 15-0
	Лимит 15-0
	+0


Краткое описание битов:
A — бит доступа (устанавливаем 0);
R — исполнение и чтение / только исполнение (1 / 0);

C — конформный / неконформный сегмент кода (устанавливаем 0);
бит 11 = 1 — сегмент кода, а не данных;

бит 12 = 1 — несистемный дескриптор;
DPL — уровень привилегий (устанавливаем наивысший 00);
P — бит присутствия (устанавливаем 1);

AVL — доступен программному обеспечению;
бит 21 зарезервирован в защищенном режиме и устанавливается в 0;
D — операнды и адресация по умолчанию 32 / 16-разрядные (1 / 0);
G — бит гранулярности: лимит масштабируется / не масштабируется (1 / 0).
В длинном режиме дескрипторы кода по-прежнему занимают 8 байт (системные дескрипторы сегментов состояния задачи, таблиц локальных дескрипторов и шлюзов вызова расширены до 16 байт). Их структура такова:

	31               24
	23
	22
	21
	20
	19                  16
	15
	14  13
	12
	11
	10
	9
	8
	7                   0
	

	Базовый

адрес 31-24
	G
	D
	L
	A

V

L
	Лимит 19-16
	P
	DPL
	1
	1
	C
	R
	A
	Базовый

адрес 23-16
	+4

	Базовый адрес 15-0
	Лимит 15-0
	+0


Серым цветом указаны поля, которые игнорируются в 64-битном режиме. Базовый адрес равен 0h вне зависимости от значения соответствующего поля. Наиболее важен бит L. Если он равен нулю, то процессор будет работать в режиме совместимости и выполнять либо 32-разрядный код (при D = 1), либо 16-разрядный (при D = 0). Если же L = 1, то процессор будет работать в 64-битном режиме. В этом случае бит D должен быть установлен в ноль. При L = 1, D = 1 возникает исключение общей защиты.
dseg

segment use16

gdt

dd 0,0

В начале таблицы располагается неиспользуемый дескриптор.
entry1
dw 0ffffh



dw 0



db 0



db 10011000b



db 01001111b



db 0

Далее размещаем дескриптор 32-разрядного кода (D = 1) для защищенного режима и режима совместимости, которому соответствует селектор 8h. При этом предполагается, что код можно только исполнять, но не читать (R = 0), и что базовый адрес сегмента 0h, а лимит FFFFFh (1 мегабайт).
entry2
dw 0



dw 0



db 0



db 10011000b



db 00100000b



db 0

Дескриптору кода для 64-битного режима соответствует селектор 10h. Бит L установлен!

entry3
dw 0ffffh

lowbase
dw ?

middbase
db ?



db 10011010b



db 0



db 0
Последний в таблице дескриптор, которому соответствует селектор 18h, описывает 16-разрядный код (D = 0) и предназначен для перехода из защищенного режима в реальный режим. Поэтому код можно исполнять и читать (R = 1), лимит выбирается FFFFh (64 кБ), что характерно для любых сегментов в реальном режиме, а в поля lowbase и middbase базы при инициализации вносится линейный адрес сегмента cseg.

lengdt
equ $-gdt

Константа lengdt — это размер таблицы глобальных дескрипторов.
pdescr
label fword



dw lengdt-1

addrgdt
dd ?

Значение переменной типа FWORD, помеченной как pdescr, используется при загрузке регистра GDTR.

info

db 'Быстрый перевод процессора в 64-битный режим и обратно'



db 13,10,13,10,'$'

notsupport
db 'Режим не поддерживается данным процессором$'

bootmess
db 'Требуется запуск MS-DOS$'

press

db 13,10,13,10,'Нажмите любую клавишу...$'

dseg

ends

Данные сообщения выводятся на экран при работе ветви er программы. Используется кодировка MS-DOS, а каждое из сообщений заканчивается символом $, как это требуется при выполнении функции вывода 9h прерывания 21h.
end

start

:exit
Точка входа обозначается как start и находится в сегменте cseg. Метка exit используется только командным интерпретатором при завершении его работы.
Вне всяких сомнений, с кодом программы можно экспериментировать, изменяя инструкции сегмента cseg64 или модифицируя системные структуры и вводя новые, например, добавляя таблицу дескрипторов прерываний длинного режима. Желательно при этом руководствоваться пословицей «Семь раз примерь, один раз отрежь», избегая лишних перезагрузок компьютера.
